



CONTENTS

=====

	Page
Introduction.....	2
Music.....	3
Adventure.....	5
Gobbler.....	8
Starbase.....	10
Candy Kid.....	15
Scroll Routine.....	22
Bits N Pieces.....	23

=====

All rights reserved.

No part of this publication may be reproduced, in part or whole without the written consent from Trident Technological Systems.

To store your programs forever! Plug your Data Cassette into your SEGA and then type SAVE"program name" and press (CR).

To LOAD your programs after you have saved them just type in LOAD"program name" and press (CR).

INTRODUCTION

This book is the first of a series of Ultimate Programs Books for the SEGA and the programs that are in it were kindly donated by two reputable SEGA programmers. They are totally original programs and are of extremely good quality. We are sure you will enjoy them as much as we enjoyed them. There is also a section on how to write your own adventures with a minimal of knowledge of computers. This part of the book explains in detail a few of the basic commands that you need to write adventures.

The Editor

MUSIC

Music, in this music program you have a great range of notes which should be enough for anyone! It shows the note you are playing and is a lot of fun for everyone!

```
10 GOTO110
20 N$=INKEY$:IFN$=""THENSOUND0:GOTO20
22 IFN$=CHR$(13)THEN250
50 X=ASC(N$)-43:IFX<0ORX>47THEN20
60 IFN(X)=1THEN20
80 SOUND1,N(X),15:SOUND2,N(X)+1,15:T=T+1:T
(T)=N(X)
90 X$=INKEY$:IFX$((">"))ANDX$(("<"))N$THEN20
100 CURSOR21,2:PRINTG$(X);" ";T$(T)=G$(X):
GOTO20
110 CLS:PRINT"PRESS CR FOR PLAY BACK!":PRI
NT:PRINT:PRINT
115 CURSOR0,2:PRINT"NOTE THAT IS PLAYING":
PRINT:PRINT"PRESS (CR) FOR PLAYBACK!"
120 DIM N(47),G$(47),T(900),T$(900)
130 FORI=1TO47:READN(I),G$(I):NEXT:GOTO20
230 DATA 220,A,1,,247,B,1,,622,D#,1,,277
231 DATA C#,311,D#,1,,370,F#,415,G#,466
232 DATA A#,1,,554,C#,1,,1,,1,,1,,1,,1,
233 DATA 698,F,1,,165,E,131,C,1,,330,E
234 DATA 139,C#,156,D#,1,,523,C,185,F#
235 DATA 208,G#,233,A#,196,G,175,F,587,D
236 DATA 659,E,262,C,349,F,117,A#,392,G
237 DATA 494,B,147,D,294,D,123,B,440,A,110
,A
```

```
250 CLS:PRINT"PRESS (CR) TO PLAY BACK":PRI
NT:PRINT"PRESS SPACE TO RETURN":CURSOR0,10
:PRINT"NOTE THAT'S PLAYING"
255 FORQ=1TOT:FORI=0TO10:NEXT:SOUND1,T(Q),
15:SOUND2,T(Q)+1,15:CURSOR22,10:PRINTT$(Q)
;" ":NEXT:SOUND0
260 N$=INKEY$:IFN$=""THEN260
270 IFN$=" "THENERASE:RESTORE:T=0:GOTO10
280 IFN$=CHR$(13)THEN250
290 GOTO260
300 FORI=0TOT:PRINTT$(I),T(I)
310 IFINKEY$=""THEN310
320 NEXT
```


ADVENTURE

Adventures are rather easy to write and you don't need a lot of experience, all you need is a little imagination. We will start with a small one location adventure and then progress to something a little more sophisticated.

For someone to convert this into a multi location adventure you could use this same format for all your locations. The only thing you would need to change would be line 50, instead of GOSUB 100 you could have GOTO (line number of your next location).

```
5 CLS:COLOR 1,5
10 PRINT "You are in a cave."
20 PRINT "Exits are; N."
30 PRINT "Visible Objects; Rock."
40 INPUT "WHAT WILL YOU DO ";A$
50 IF A$ = "N" THEN GOSUB 100
60 IF A$ = "GET ROCK" THEN GOSUB 200
70 IF A$ = "SIT ROCK" THEN GOSUB 300
80 IF A$ = "MOVE ROCK" THEN GOSUB 400
90 BEEP:GOTO 5
100 PRINT "THERE IS A ROCK IN THE WAY!":GOTO
500
200 PRINT "SORRY IT'S TOO HEAVY!":GOTO 500
300 PRINT "AHHH THATS BETTER!":GOTO 500
400 PRINT "SORRY IT'S TOO HEAVY!":GOTO500
500 FOR A = 0 TO 500:NEXT:RETURN
```

LINE 5: CLS means clear screen and it does just that. COLOR 1,5 puts black writing on blue screen. Refer to the Basic Manual for more color combinations.

LINE 10: This is just an ordinary PRINT statement and will print anything inside the quotation marks. After going to the end of the line it then goes to the next line unless there was another directing it somewhere else.

LINE 20: As with 10.

LINE 30: As with 10.

LINE 40: Now INPUT is similar to PRINT except that after it has printed it waits for you to input something (type it in and then press CR) and then it stores it in A\$ until A\$ has been declared as equaling something else.

LINE 50: IF, this statement asks the question. If it doesn't comply then moves to the next line. If it does comply THEN it does what ever comes after THEN. GOSUB sends the computer to a designated line and on the command RETURN sends it back to where it left. So if there wasn't anything after the GOSUB statement then it would proceed to the next line.

LINE 60: As with 50

LINE 70: As with 50

LINE 80: As with 50

LINE 90: BEEP makes a BEEP. GOTO sends the computer to the designated line.

LINE 100: As with line 10 and line 90

LINE 200: As with 100

Line 300: As with 100

LINE 400: As with 100

LINE 500: FOR makes a variable equal from one amount TO another amount. Now a variable can only equal one thing at a time so what happens is that the variable will equal (in this case 0) 0 until it gets to the NEXT command. Now what the NEXT command does is it sends it back to the FOR and adds 1 to the value. It does this until it has reached (in this case 500) 500 then it continues. The RETURN does as mentioned in line 50.

GOBBLER

Gobbler, this is the smallest arcade game there is in the book but is thoroughly enjoyable thanks to the quality of the routines in this program. You must use the four cursor keys for movement so you can gobble up the dots but beware of the poison apples that will kill you straight away!! Excellent value for size!!

```
10 CLS:PATTERN#253,"3030CCB4B43048CC"
20 COLOR1,3:A=15443:CURSOR1,0:PRINT"SCORE"
30 CURSOR13,0:PRINT"HI-SCORE";HS
40 CURSOR29,0:PRINT"GOBBLER"
50 FORZ=1TO21:CURSOR0,Z
60 PRINT"■.....■"
70 NEXT
80 FORZ=0TO36:CURSORZ,1
90 PRINT"■":CURSORZ,21:PRINT"■":NEXT:GOTO130
100 B$="●":IFRND(1)*2>1THENB$="."
110 CURSORRND(1)*35+1,RND(1)*19+2
120 PRINTB$:A$=INKEY$:IFA$="" THEN 100
130 L=(A$=CHR$(29))-(A$=CHR$(28))+((A$=CHR$(30))
-(A$=CHR$(31)))*40
```



```
140 I=VPEEK(A+L):IFI<165THEN240
150 IFI=229THEN100
160 IFI=236THENFORZ=1TO100:BEEP1
170 BEEP0:NEXT:CLS:PRINT"Score";S:GOTO190
180 A=A+L:VPOKEA-L,32:VPOKEA,253:GOTO100
190 IFHS<STHENHS=S:S=0
200 CURSOR5,21:PRINT"PRESS <SPACE TO CONT>"
210 IFINKEY$=""THEN210
220 IFINKEY$=" "THENBEEP:GOTO10
230 FORI=0TO50:BEEP1:BEEP0:NEXTI:GOTO210
240 IFI=32THEN180
250 BEEP1:BEEP0:S=S+1:CURSOR7,0:PRINTS:GOTO180
```

STARBASE

In STARBASE you must defend your space station from attacking Alien UFO's!! The game starts easy and will slowly increase in difficulty until you will no longer be able to stop the aliens from destroying your station. Use the arrow keys to point in the direction you wish to fire your lasers. To fire press the space bar.

```
10 CLS:SCREEN 2,2:COLOR 15,1,,1:MAG 3:GOSUB 460
:GOSUB 700
20 A$=INKEY$:IFA$>" "ORA$=""THEN100
30 IFA$=" "THENGOSUB120:GOTO100
40 ON ASC(A$)-27 GOTO 50,60,70,80
50 D=2:GOTO 90
60 D=4:GOTO 90
70 D=1:GOTO 90
80 D=3
90 SPRITE1,(80,80),INT(4*(D-1)),15:GOTO 100
100 DL=DL-.002:GOSUB 330:GOSUB 360
110 GOTO 20
120 ON D GOTO 130,180,230,280
130 XX=UY(1):IF UX(1)=0 THEN XX=22
140 BEEP 1:LINE(95,82)-(96,XX+2),8,8F
```

```

150 LINE(95,12)-(96,82),1,BF:BEEP 0:IF UX(1)=0
THEN 450

160 SOUND 4,2,15:FOR Z=5 TO 0STEP-1:SPRITE3,(UX
(1),UY(1)),20,Z:NEXTZ:UX(1)=0:SOUND0:GOSUB 430

170 RETURN

180 XX=UX(2):IF XX=0THEN XX=162

190 BEEP 1:LINE(112,96)-(XX+2,97),8,BF

200 LINE(112,96)-(162,97),1,BF:BEEP 0:IF UX(2)=
0 THEN 450

210 SOUND 4,2,15:FOR Z=5 TO 0STEP-1:SPRITE4,(UX
(2),UY(2)),20,Z:NEXTZ:UX(2)=0:SOUND0:GOSUB 430

220 RETURN

230 XX=UY(3):IF UX(3)=0 THEN XX=172

240 BEEP 1:LINE(95,112)-(96,XX+2),8,BF

250 LINE(95,112)-(96,176),1,BF:BEEP 0:IF UX(3)=
0 THEN 450

260 SOUND 4,2,15:FOR Z=5 TO 0STEP-1:SPRITE5,(UX
(3),UY(3)),20,Z:NEXTZ:UX(3)=0:SOUND0:GOSUB 430

270 RETURN

280 XX=UX(4):IF XX=0 THEN XX=32

290 BEEP 1:LINE(XX+2,96)-(82,97),8,BF

300 LINE(22,96)-(82,97),1,BF:BEEP 0:IF UX(4)=0
THEN 450

310 SOUND 4,2,15:FOR Z=5 TO 0STEP-1:SPRITE6,(UX
(4),UY(4)),20,Z:NEXTZ:UX(4)=0:SOUND0:GOSUB 430

```

```

320 RETURN

330 IF RND(1)<DL THEN RETURN

340 FOR Z=1 TO 4: IF UX(Z)=0 AND RND(1)>DL THEN UX(Z)=UA
(Z): UY(Z)=UB(Z): RETURN

350 NEXT Z: RETURN

360 FOR Z=1 TO 4: IF UX(Z)=0 THEN NEXT Z: RETURN

370 BEEP1: BEEP0: UX(Z)=UX(Z)+U1(Z): UY(Z)=UY(Z)+U
2(Z): SPRITE Z+2,(UX(Z),UY(Z)),20,6

380 IF UY(1)=54 OR UY(3)=108 OR UX(2)=110 OR UX
(4)=50 THEN 400

390 NEXT Z: RETURN

400 SPRITE1,(0,0),0,0: SOUND4,2,15: FOR Z=15 TO 0
STEP -.4: SPRITE 0,(80,80),16,RND(1)*15: SOUND4,2
,Z: NEXT Z: SOUND0: SPRITE0,(80,80),16,0

410 IF SC>HS THEN HS=SC: BEEP: SCREEN 1,1: CLS: PRI
NT"YOU HAVE THE NEW HIGH SCORE": PRINT: PRINT: PRI
NT"SCORE:";HS

415 BEEP: FOR Z=1000 TO 140 STEP -20: SOUND1,Z,15
: NEXT Z: SOUND0: FOR Z=1 TO 800: NEXT Z: CLS: SCREEN2,2

420 CLS: GOTO 10

430 SC=SC+10

440 BLINE(186,30)-(250,38),,BF: CURSOR 188,30: CO
LOR 15: PRINT SC: RETURN

450 SC=SC-10: GOTO 440

460 CLS: CURSOR 56,10: PRINT CHR$(17);

```

```
470 PRINT"* STARBASE *";CHR$(16)
480 CURSOR20,70:PRINT"LAST SCORE:";SC:PRINT"
:PRINT"      HI-SCORE:";HS
490 CURSOR 52,175:COLOR 2
500 PRINT"PRESS <S> TO BEGIN PLAYING"
510 IFINKEY$(<)"S"THEN510
520 CLS:GOSUB530:RETURN
530 PATTERN S#0,"0001010100000000"
540 PATTERN S#2,"0080808000000000"
550 PATTERN S#6,"0000000000000000E"
560 PATTERN S#7,"0E00000000000000"
570 PATTERN S#9,"0000000001010100"
580 PATTERN S#11,"0000000080808000"
590 PATTERN S#12,"00000000000000070"
600 PATTERN S#13,"7000000000000000"
610 PATTERN S#16,"01020602072C7A89"
620 PATTERN S#17,"897A2C0702060201"
630 PATTERN S#18,"80406040E0345E91"
640 PATTERN S#19,"915E34E040604080"
650 PATTERN S#20,"010205090B3F55BF"
660 PATTERN S#21,"4A3F0B0905020100"
```



```
670 PATTERN S#22,"8040A090D0FC52FD"  
680 PATTERN S#23,"AAFCD090A0408000"  
690 RETURN  
700 D=1:DL=.8:SC=0:CURSOR188,20:COLOR 3:PRINT "  
SCORE:"  
710 FORZ=1TO4:UX(Z)=0:UY(Z)=0:NEXTZ  
720 UA(1)=80:UB(1)=10:U2(1)=4  
730 UA(2)=150:UB(2)=80:U1(2)=-4  
740 UA(3)=80:UB(3)=160:U2(3)=-4  
750 UA(4)=10:UB(4)=80:U1(4)=4  
760 SPRITE0,(80,80),16,7:LINE(180,10)-(255,50),  
15,8  
770 RETURN
```

CANDY KID

Candy Kid is an arcade style game where you must go around the screen eating all the candy without running into a Candy Devil or without being caught by the Candy Mama. There are several different screens and you use the four cursor keys for movement. VERY ADDICTIVE!!

```
10 MAG1:GOSUB 650:HS=0:GOTO 760
20 SCREEN 2,2:CLS:GOSUB 260
30 A$=INKEY$
40 IF A$=CHR$(28) AND CX<170 THEN CX=CX+16
50 IF A$=CHR$(29) AND CX>10 THEN CX=CX-16
60 IF A$=CHR$(30) AND CY>5 THEN CY=CY-16
70 IF A$=CHR$(31) AND CY<165 THEN CY=CY+16
80 VY=CY+4:A=VPEEK(INT(VY/8)*256+INT(CX/8)
*8+VYMOD8)
90 SPRITE3,(CX,CY),0,2
100 IF A=3 THEN BEEP1:BEEP0:CA=CA+1:SC=SC+
10:BLINE(CX,CY)-(CX+11,CY+11),,BF:GOSUB220
110 IF A=19 THEN CK=CK-1:GOSUB230:IFCK=0TH
EN910
120 IF BX>CX THEN BX=BX-8:GOTO 150
130 IF BX<CX THEN BX=BX+8:GOTO 150
140 BX=CX
150 IF BY>CY THEN BY=BY-8:GOTO 180
```

```

160 IF BY<CY THEN BY=BY+8:GOTO 180

170 BY=CY

180 SPRITE1,(BX,BY),8,5

190 IF BX>CX-12AND BX<CX+12ANDBY>CY-12ANDB
Y<CY+12 THENCK=CK-1:GOSUB230:IFCK=0THEN910

200 IF CA=CS THEN GOTO 240

210 GOTO 30

220 BLINE (195,50)-(225,58),15,BF:CURSOR19
5,50:COLOR15:PRINT SC:RETURN

230 GOSUB 580:CURSOR195,120:COLOR4:PRINT "
WATCH OUT!":FOR I=0 TO 250:NEXT I:CX=10:
CY=5:BX=170:BY=165:GOSUB 1000:RETURN

240 BLINE(195,120)-(255,128),,BF:GOSUB 510
:SC=SC+ST*300:GOSUB 220:CURSOR200,120:PRIN
T " NICE !"

250 FOR I=0 TO 250:NEXT I:ST=ST+1:SS=SS+
1:GOSUB 1020:GOTO 20

260 LINE (7,2)-(191,185),10,B:FOR X=14 TO
174 STEP 16:FOR Y=9 TO 169 STEP 16

270 C=INT(RND(1)*12)+2:COLORC:CURSORX,Y:PR
INT "#":NEXT Y:NEXT X

280 IF SS=4 THEN SS=1

290 IF SS=1 THEN CS=96:GOTO 320

300 IF SS=2 THEN CS=98:GOTO 340

310 IF SS=3 THEN CS=85:GOTO 370

```

```
320 FOR X=26 TO 154 STEP 32:FOR Y=21 TO 14
9 STEP 32:GOSUB 440:NEXT Y:NEXT X

330 GOSUB 450:GOSUB 510:RETURN

340 FOR X=26 TO 154 STEP 64:FOR Y=21 TO 14
9 STEP 32:GOSUB 440:NEXT Y:NEXT X

350 FOR X=58 TO 122 STEP 64:FOR Y=37 TO 13
3 STEP 32:GOSUB 440:NEXT Y:NEXT X

360 GOSUB 450:GOSUB 510:RETURN

370 RESTORE730

380 READ X

390 READ Y

400 IF Y=-1 THEN 380

410 IF Y=-2 THEN 430

420 GOSUB 440:GOTO 390

430 GOSUB 450:GOSUB 510:RETURN

440 COLOR8 :CURSORX,Y:PRINT "$":CURSORX,Y+
8:PRINT "%":CURSORX+8,Y:PRINT "&":CURSORX+
8,Y+8:PRINT "'":RETURN

450 CURSOR195,15:COLOR9:PRINT "HI SCORE":C
URSOR195,25:PRINT HS

460 CURSOR195,40:COLOR10:PRINT "SCORE":CUR
SOR195,50:PRINT SC

470 GOSUB 1000

480 CURSOR195,90:COLOR5:PRINT "STAGE":CURS
OR195,100:PRINT ST
```

```

490 FOR C=1 TO 13:COLORC:CURSOR200,150:PRINT
"*****":CURSOR200,158:PRINT "*CANDY*"
:CURSOR200,166:PRINT"* KID *":CURSOR200,17
4:PRINT"*****":NEXT C

500 RETURN

510 RESTORE 740

520 READ T1

530 READ S1

540 IF S1=-1 THEN GOTO 520

550 IF S1=-2 THEN SOUND0:RETURN

560 SOUND1,S1,15:FOR I=0 TO T1:NEXT I

570 GOTO 530

580 RESTORE 750

590 READ T2

600 READ S2

610 IF S2=-1 THEN GOTO 590

620 IF S2=-2 THEN RETURN

630 SOUND1,S2,15: FOR I=0 TO T2:NEXT I:SOUND0

640 GOTO 600

650 FOR I=0 TO 15:READ Q$:PATTERNS#I,Q$:N
EXT I

660 FOR I=35 TO 39:READ Q$:PATTERN#I,Q$:N
EXT I:RETURN

```


670 DATA 030F1F3F33616173,7F33381C8F53A050
,C0F0F8FCCC8686CE,FECC1C38F1CA050A

680 DATA 00000000000C0C00,000C070300804020
,0000000000303000,0030E0C000010204

690 DATA 03070F1F193070F9,FE9C0F0F0F3F7F73
,C0E0F0F8980C0E9F,7F39F0F0F0FCFECE

700 DATA 0000000000060600,0003101010000000
,0000000000606000,00C0080808000000

710 DATA C0C03C3C3C3C0303

720 DATA 100F45E543424161,62524A444444040F
,08F0A0A0E0408080,444E4424243820F0

730 DATA 26,21,37,53,85,117,133,149,-1,42,
21,149,-1,58,21,53,69,101,117,149,-1,74,53
,117,-1,90,21

735 DATA 149,-1,106,53,117,-1,122,21,53,69
,101,117,149,-1,138,21,149,-1,154,21,37,53
,85,117,133,149,-2

740 DATA 20,523,659,784,659,784,659,-1,60,
784,-2

750 DATA 40,523,-1,20,523,523,-1,40,523,-1
,20,622,587,587,523,523,494,-1,60,523,-2

760 SCREEN 2,2:COLOR1,1,,1:COLOR8:CLS

770 FOR C=3 TO 9:COLORD

780 CURSOR58,20:PRINT"*****"
***"

790 CURSOR72,28:PRINTCHR\$(17);"CANDY KID"

```
800 CURSOR58,36:PRINTCHR$(16);"*****  
*****"
```

```
810 NEXTC
```

```
820 CURSOR58,108:COLOR2:PRINT ". . . . .  
.CANDY KID"
```

```
830 CURSOR58,124:COLOR5:PRINT ". . . . .  
.CANDY MAMA"
```

```
840 X=42:Y=137:COLOR10:CURSORX,Y:PRINT "$"  
:CURSORX,Y+8:PRINT "%":CURSORX+8,Y:PRINT "  
&":CURSORX+8,Y+8:PRINT "'"
```

```
850 CURSOR58,140:PRINT ". . . . .THE DE  
VIL"
```

```
860 CURSOR58,156:PRINT ". . . . .CANDY  
10 PTS "
```

```
870 FOR X1=0 TO 42 STEP 2:SPRITE3,(X1,101)  
,0,2:SPRITE2,(X1,101),4,11:SPRITE1,(X1,117)  
,8,5:SPRITE0,(X1,117),12,6:NEXT X1
```

```
880 COLOR2:CURSOR46,157:PRINT "#":BEEP:FOR  
I=0 TO 50:NEXT I
```

```
890 GOSUB 510:CURSOR60,180:COLOR10:PRINT "  
Please push [CR] key"
```

```
900 Y#=INKEY#:IF NOT Y#=CHR$(13) THEN BEEP  
1:BEEP0:GOTO900
```

```
905 GOSUB 1010:GOSUB 1020:GOTO20
```

```
910 CLS:IF SC>HS THEN HS=SC
```

```
920 CURSOR100,40:COLOR9:PRINT "GAME OVER"
```

```
930 CURSOR90,70:COLOR13:PRINT"HI-SCORE";HS
```

```
940 CURSOR90,90:COLOR10:PRINT "YOUR SCORE"  
;SC
```

```
950 CURSOR50,120:COLOR2:PRINT "One more tr  
y - Push [CR] key"
```

```
960 CURSOR50,140:COLOR5:PRINT "No thank yo  
u - Push [E] key"
```

```
970 Y#=INKEY#:IF NOT Y#=CHR$(13) THEN BEEP  
1:BEEP0:GOTO970
```

```
975 IFY#=CHR$(13)THENGOSUB 1010:GOSUB 1020  
:GOTO 20
```

```
980 IF Y#="E" THEN SCREEN1,1:COLOR1,1:END
```

```
990 GOTO 970
```

```
1000 CURSOR195,65:COLOR2:PRINT "KID":BLINE  
(200,75)-(205,83),,BF:CURSOR195,75:PRINT C  
K:RETURN
```

```
1010 SC=0:CK=3:ST=1:SS=1:RETURN
```

```
1020 CX=10:CY=5:BX=170:BY=165:CA=0:RETURN
```

SCROLL

To Scroll the screen down CALL &H9808 anytime after running this program. Write your own program around this scroll routine by using the VPEEK and VPOKE commands.

```
1 REMXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX  
XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX  
XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX  
XXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXXX
```

```
2 FORZ=&H9808TO&H9854:READA#:POKEZ,VAL  
("&H"+A#):NEXTZ
```

```
3 DATA F3,21,59,3F,06,17,11,27
```

```
4 DATA 00,C5,E5,CD,48,98,06,28
```

```
5 DATA 21,63,98,DB,BE,77,23,10
```

```
6 DATA FA,E1,E5,19,CD,48,98,21
```

```
7 DATA 63,98,06,28,7E,D3,BE,23
```

```
8 DATA 10,FA,E1,B7,ED,52,C1,10
```

```
9 DATA D8,21,FF,38,06,28,CD,48
```

```
10 DATA 98,3E,00,D3,BE,10,FA,C9
```

```
11 DATA 7D,D3,BF,7C,E6,3F,F6,40
```

```
12 DATA D3,BF,DB,BE,C9
```

BITS & PIECES

5 REM CHANGES LOWERCASE LETTERS TO UPPER
CASE REVERSED

10 FORZ=97TO122:PRINTCHR\$(Z);:NEXT

20 FOR Z=&H1A08 TO &H1AD7:VPOKE Z+&H100,VP
EEK(Z) XOR &HFF:NEXT

10 REM REVERSES ALL CHARACTERS

20 FORZ=32TO255:PRINTCHR\$(Z);:NEXT

30 FOR Z=&H1900 TO &H1FFF:VPOKE Z,VPEEK(Z
) XOR &HFF:NEXT Z

10 REM ENLARGES ALL CHARACTERS

20 FORZ=32TO255:PRINTCHR\$(Z);:NEXT

30 FOR Z=&H1900 TO &H1FFF:VPOKE Z,VPEEK(Z)
OR VPEEK(Z)/2:NEXT Z

INPUT ROUTINE: to use routine
GOSUB 950 On pressing the (CR) key A\$
will contain what was typed in.
DEL key backspaces.
to position X=x axis, Y=y axis

```
10 REM Sample program using input routine
lines 10 to 60

20 CLS:PRINT"WHAT IS YOUR NAME ?"

30 X=0:Y=2:GOSUB 950

40 PRINT:PRINT:PRINT"HELLO ";A$

50 FOR Z=1 TO 300:BEEP 1:BEEP 0:NEXT Z

60 GOTO 20

950 A$=""

951 CURSOR X,Y:PRINT CHR$(143);

952 B$=INKEY$:IF B$="" THEN 952

953 A=ASC(B$):IF A=8 AND LEN(A$)>0 THEN 958

954 IF A=13 THEN BEEP:CURSOR X,Y:PRINT " ";
:RETURN

955 IF A<32 OR A>122 THEN 952

956 IF X>36THEN 952

957 BEEP1:A$=A$+B$:CURSOR X,Y:PRINT B$;;X=X
+1:BEEP 0:GOTO 951

958 BEEP:A$=LEFT$(A$,LEN(A$)-1):X=X-1:CURSOR
X,Y:PRINT " ";:GOTO 951
```



